Use of the Disk II Interface Card Through Your Own Software

by John Uhley

This is the first of a series of three articles which will discuss several techniques used in accessing the Disk II Interface Card. The routines discussed in these articles will enable a programmer to access the Disk II without the use of Apple DOS or Apple RWTS. This article will introduce the programmer to one method of accessing the Disk II Interface card and develop several routines to position the disk drive's magnetic head across the surface of the diskette.

Each of the eight slots in back of the Apple computer is allocated 16 memory locations for I/O control. Some of these memory locations act as softswitches (software switches) and perform a predefined hardware task when addressed by a software routine. Other I/O memory locations are used as wormholes through which data can enter or exit the computer.

Most of the Apple's I/O is done on page $C0 of memory. The following table illustrates the range of memory reserved for each slot's softswitches.

SLOT	LOCATIONS
0	$C080 - $C08F
1	$C090 - $C09F
2	$C0A0 - $C0AF
3	$C0B0 - $C0BF
4	$C0C0 - $C0CF
5	$C0D0 - $C0DF
6	$C0E0 - $C0EF
7	$C0F0 - $C0FF

One common method of accessing softswitches through software is to use the Apple's indexed addressing mode. By adding various values to the address of a slot zero softswitch it is possible to change the addressed slot by altering the index register alone. For example, if a program wanted to access the softswitches of the card in Slot 3 the following method could be used:

```
        LDX   #$30      ; (select slot 3)

TURNON  LDA   $C089,X   ; turn on disk drive in slot 3
        RTS             ; and return
```

Using this technique the same routine can access any slot's softswitches by changing the value in the X-register:

```
        LDX   #$60      ; (select slot 6)
        JMP   TURNON    ; turn on disk drive in slot 6
                        ; using the same routine
```

In the table below 'S' represents one of the 16 softswitches of the specified slot.

SLOT	SOFTSWITCH	X-REG	COMMAND
0	S	$00	LDA $C08S,X
1	S	$10	LDA $C08S,X
2	S	$20	LDA $C08S,X
3	S	$30	LDA $C08S,X
4	S	$40	LDA $C08S,X
5	S	$50	LDA $C08S,X
6	S	$60	LDA $C08S,X
7	S	$70	LDA $C08S,X

Depending on the actual card placed in a given slot the 16 softswitches will perform a different function. This article will only discuss the effect of these 16 softswitches on the Disk II Interface Card. The table below illustrates the functions of each of the softswitches.

MEMORY LOCATION	FUNCTION
$C080 + SLOT	Phase 0 off (positioning)
$C081 + SLOT	Phase 0 on (positioning)
$C082 + SLOT	Phase 1 off (positioning)
$C083 + SLOT	Phase 1 on (positioning)
$C084 + SLOT	Phase 2 off (positioning)
$C085 + SLOT	Phase 2 on (positioning)
$C086 + SLOT	Phase 3 off (positioning)
$C087 − SLOT	Phase 3 on (positioning)
$C088 + SLOT	Power Down (drive off)
$C089 - SLOT	Power Up (drive on)
$C08A + SLOT	Select 1 (select drive 1)
$C08B + SLOT	Select 2 (select drive 2)
$C08C + SLOT	Readswitch (I/O wormhole)
$C08D + SLOT	Writeswitch (I/O wormhole)
$C08E + SLOT	Clearswitch (I/O wormhole)
$C08F + SLOT	Shiftswitch (I/O wormhole)

(SLOT refers to the index value needed to access the softswitches of a given slot. As shown earlier, this value is equal to that slot number times 16.)

The following routines demonstrate some of the techniques used to activate or deactivate a disk drive. Each time a new disk

drive is activated it is necessary to wait for the drive's motor to reach operational speed. One suitable delay loop is illustrated by the "MWAIT" subroutine.

The first eight softswitches are used to position the disk drive's magnetic head above the physical tracks of a diskette. These softswitches are used to rotate a motor which moves the magnetic head back and forth along the surface of a diskette.

By rotating the motor in a clockwise direction the magnetic head is moved towards higher numbered tracks. Conversely, counterclockwise rotation forces the magnetic head towards lower numbered tracks. Figures A and B illustrate the concepts discussed in these paragraphs using simplified models.

Figure C shows a magnetized needle surrounded by four electromagnetic poles. By magnetizing one of the four poles the needle is forced to 'point' towards that pole. By magnetizing and demagnetizing the poles in a given order the needle can be made to spin in a clockwise or counterclockwise direction (see figure D).

By replacing the needle with a motor and the poles with software controlled electromagnets (numbered 0,1,2,and 3) a model of the disk drive's positioning motor can be visualized (see figure E).

```
                        ********************************
                        *                              *
                        * ROUTINE TO TURN ON DRIVE 1   *
                        *                              *
                        ********************************
                SLOT    EQU   $0001
                WAIT    EQU   $0002
                DISKON  EQU   $C089
                DRIVEA  EQU   $C08A
                        ********************************
                        *                              *
                        * SET SLOT EQUAL TO SLOT 6     *
                        *                              *
                        ********************************
7000: A9 60             LDA   #$60
7002: 85 01             STA   SLOT
                        ********************************
                        *                              *
                        * TURN ON THE DISK DRIVE AND   *
                        * SELECT DRIVE A               *
                        *                              *
                        ********************************
7004: A6 01             LDX   SLOT
7006: BD 89 C0          LDA   DISKON,X
7009: BD 8A C0          LDA   DRIVEA,X
                        ********************************
                        *                              *
                        * WAIT FOR DRIVE TO POWER UP   *
                        *                              *
                        ********************************
700C: A9 EF     MWAIT   LDA   #$EF
700E: 85 02             STA   WAIT
7010: A9 D8             LDA   #$D8
7012: 85 03             STA   WAIT+1
7014: A0 12     MWAITA  LDY   #$12
7016: 88        MWAITB  DEY
7017: D0 FD             BNE   MWAITB
7019: E6 02             INC   WAIT
701B: D0 F7             BNE   MWAITA
701D: E6 03             INC   WAIT+1
701F: D0 F3             BNE   MWAITA
7021: 60                RTS
```

```
                        ********************************
                        *                              *
                        * ROUTINE TO TURN ON DRIVE 2   *
                        *                              *
                        ********************************
                SLOT    EQU   $0001
                WAIT    EQU   $0002
                DISKON  EQU   $C089
                DRIVEB  EQU   $C08B
                        ********************************
                        *                              *
                        * SET SLOT EQUAL TO SLOT 6     *
                        *                              *
                        ********************************
7000: A9 60             LDA   #$60
7002: 85 01             STA   SLOT
                        ********************************
                        *                              *
                        * TURN ON THE DISK DRIVE AND   *
                        * SELECT DRIVE B               *
                        *                              *
                        ********************************
7004: A6 01             LDX   SLOT
7006: BD 89 C0          LDA   DISKON,X
7009: BD 8B C0          LDA   DRIVEB,X
                        ********************************
                        *                              *
                        * WAIT FOR DRIVE TO POWER UP   *
                        *                              *
                        ********************************
700C: A9 EF     MWAIT   LDA   #$EF
700E: 85 02             STA   WAIT
7010: A9 D8             LDA   #$D8
7012: 85 03             STA   WAIT+1
7014: A0 12     MWAITA  LDY   #$12
7016: 88        MWAITB  DEY
7017: D0 FD             BNE   MWAITB
7019: E6 02             INC   WAIT
701B: D0 F7             BNE   MWAITA
701D: E6 03             INC   WAIT+1
701F: D0 F3             BNE   MWAITA
7021: 60                RTS
```

```
                        ********************************
                        *                              *
                        * ROUTINE TO TURN OFF A DISK   *
                        *                              *
                        ********************************
                SLOT    EQU   $0001
                DISKOFF EQU   $C088
                        ********************************
                        *                              *
                        * SET SLOT EQUAL TO SLOT 6     *
                        *                              *
                        ********************************
7000: A9 60             LDA   #$60
7002: 85 01             STA   SLOT
                        ********************************
                        *                              *
                        * TURN OFF THE DISK DRIVE      *
                        * AND QUIT                     *
                        *                              *
                        ********************************
7004: A6 01             LDX   SLOT
7006: BD 88 C0          LDA   DISKOFF,X
7009: 60                RTS
```



Figure A



Figure C



Figure B



Figure D



Figure E

Each of the eight softswitches used in positioning the magnetic head actually activates or deactivates a specific phase of the positioning motor. By addressing these softswitches in a specific order the motor can be rotated and the magnetic head positioned back and forth along the surface of a diskette.

The following table illustrates the function of each of the eight 'positioning' softswitches. Note that SLOT refers to the index value needed to access the softswitches of a given slot.

MEMORY LOCATION	PHASE	FUNCTION
$C080 + SLOT	0	Turn phase 0 off
$C081 + SLOT	0	Turn phase 0 on
$C082 + SLOT	1	Turn phase 1 off
$C083 + SLOT	1	Turn phase 1 on
$C084 + SLOT	2	Turn phase 2 off
$C085 + SLOT	2	Turn phase 2 on
$C086 + SLOT	3	Turn phase 3 off
$C087 + SLOT	3	Turn phase 3 on

When positioning the magnetic head via software, it is necessary to wait for the positioning motor to physically move to an activated phase before that phase is deactivated. For example, in order to 'pulse' phase 0 of the positioning motor the following routine might be executed.

```
LSX   #$60        ; (select slot 6)
:
LDA
LDA   $C081,X     ; turn on phase 0
LDA   #$56        ; wait for motor to move
JSR   $FCA8       ; to the phase
LDA   $C080,X     ; before deactivating phase 0
:
RTS               ; and returning
```

The position of the tracks recognized by Apple DOS does not correspond to the phases of the positioning motor in a 1:1 ratio. Instead, the correspondence of DOS tracks to phases is in the ratio of 1:2. All even numbered tracks are positioned "under" phase 0 and all odd numbered tracks are positioned "under" phase 2. When positioned on phases 1 or 3 the disk drive's magnetic head is positioned over a half-track (much like a car driving in two lanes). Half-tracks are not used by Apple DOS although some protected software makes use of them.

The phase to which a DOS track corresponds can be calculated by multiplying the DOS track number by two (using the ASL opcode). This value represents the number of phases that must be pulsed from phase 0 (of track zero) in order to reach the specified DOS track.

The following routine can be used to position the disk drive's magnetic head over any physical track of a diskette. All DOS track numbers must be multiplied by two to account for the unused phases (half-tracks). Access to half tracks can be accomplished by setting the least significant bit of the multiplied track number to 1. If you don't know the current track number you must recalibrate the magnetic head to track zero using the "RECAL" routine provided.

```
                        1     ********************************
                        2     *                              *
                        3     * ROUTINE TO POSITION FROM     *
                        4     * ANY TRACK TO TRACK ZERO      *
                        5     *                              *
                        6     * RECALIBRATION ROUTINE        *
                        7     *                              *
                        8     ********************************
                        9     CURTRK   EQU   $0004
                        10    DESTRK   EQU   $0005
                        11    POSITION EQU   $0800
                        12    ********************************
                        13    *                              *
                        14    * SET UP "CURTRK" & "DESTRK"   *
                        15    * FOR THE RECALIBRATION        *
                        16    *                              *
                        17    ********************************
7000:  A9 80            18    RECAL    LDA   #$80
7002:  85 04            19             STA   CURTRK
7004:  A9 00            20             LDA   #$00
7006:  85 05            21             STA   DESTRK
                        22    ********************************
                        23    *                              *
                        24    * CALL POSITION TO DO THE      *
                        25    * DIRTY WORK...                *
                        26    *                              *
                        27    * NOTE: THE POSITION ROUTINE   *
                        28    *       IS ASSUMED TO BE       *
                        29    *       ORG'ED AND OBJ'D AT    *
                        30    *       MEMORY LOCATION $800   *
                        31    *                              *
                        32    ********************************
7008:  4C 00 08         33             JMP   POSITION

--- END ASSEMBLY ---

TOTAL ERRORS: 00

11 BYTES OF OBJECT CODE
WERE GENERATED THIS ASSEMBLY.

:
```

```
                        1     ********************************
                        2     *                              *
                        3     * ROUTINE TO POSITION FROM     *
                        4     * "CURTRK" TO "DESTRK"         *
                        5     *                              *
                        6     *                              *
                        7     * REMEMBER!!! CURTRK AND      *
                        8     *              DESTRK REFER    *
                        9     *              TO PHASES...    *
                        10    *              ACTUAL DOS      *
                        11    *              TRACKS MUST     *
                        12    *              BE MULTIPLIED   *
                        13    *              BY TWO BEFORE   *
                        14    *              USE IN THIS     *
                        15    *              ROUTINE         *
                        16    *                              *
                        17    ********************************
                        18    SLOT     EQU   $0001
                        19    WAIT     EQU   $0002
                        20    CURTRK   EQU   $0004
                        21    DESTRK   EQU   $0005
                        22    PHSOFF   EQU   $C080
                        23    PHSON    EQU   $C081
                        24    DISKON   EQU   $C089
                        25    DISKOFF  EQU   $C088
                        26    DRIVEA   EQU   $C08A
                        27    MONWAIT  EQU   $FCA8
                        28    ********************************
                        29    *                              *
                        30    * SET SLOT EQUAL TO SLOT 6     *
                        31    *                              *
                        32    ********************************
7000:  A9 60            33    POSITION LDA   #$60
7002:  85 01            34             STA   SLOT
                        35    ********************************
                        36    *                              *
                        37    * TURN ON DRIVE A AND WAIT     *
                        38    *                              *
                        39    ********************************
```

```
7004:  A6 01     40            LDX   SLOT
7006:  BD 89 C0  41            LDA   DISKON,X
7009:  BD 8A C0  42            LDA   DRIVEA,X
700C:  A9 EF     43            LDA   #$EF
700E:  85 02     44            STA   WAIT
7010:  A9 D8     45            LDA   #$D8
7012:  85 03     46            STA   WAIT+1
7014:  A0 12     47    MWAITA  LDY   #$12
7016:  88        48    MWAITB  DEY
7017:  D0 FD     49            BNE   MWAITB
7019:  E6 02     50            INC   WAIT
701B:  D0 F7     51            BNE   MWAITA
701D:  E6 03     52            INC   WAIT+1
701F:  D0 F3     53            BNE   MWAITA
                 54    ********************************
                 55    *                              *
                 56    * CHECK & BRANCH ON CURTRK-     *
                 57    * DESTRK RELATIONSHIP          *
                 58    *                              *
                 59    ********************************
7021:  A5 04     60    MAINLOOP LDA   CURTRK
7023:  C5 05     61             CMP   DESTRK
7025:  F0 21     62             BEQ   ALLDONE
7027:  90 07     63             BCC   MOVEUP
7029:  B0 00     64             BCS   MOVEDOWN
                 65    ********************************
                 66    *                              *
                 67    * MOVE DOWN TO LAST PHASE       *
                 68    *                              *
                 69    ********************************
702B:  C6 04     70    MOVEDOWN DEC   CURTRK
702D:  4C 32 70  71             JMP   DOWORK
                 72    ********************************
                 73    *                              *
                 74    * MOVE UP TO NEXT PHASE         *
                 75    *                              *
                 76    ********************************
7030:  E6 04     77    MOVEUP   INC   CURTRK
                 78    ********************************
                 79    *                              *
                 80    * COMPUTE PHASE NUMBER          *
                 81    * FROM THE "NEW" CURTRK         *
                 82    *                              *
                 83    ********************************
7032:  A5 04     84    DOWORK   LDA   CURTRK
7034:  29 03     85             AND   #$03 ;0-3 PHS
7036:  0A        86             ASL   ;$0-$7
                 87    ********************************
                 88    *                              *
                 89    * GET INDEXING FOR CUR SLOT#    *
                 90    *                              *
                 91    ********************************
7037:  05 01     92             ORA   SLOT
7039:  A8        93             TAY
                 94    ********************************
                 95    *                              *
                 96    * TURN ON PHASE TO MOVE &       *
                 97    * WAIT FOR PHYSICAL ACTION      *
                 98    *                              *
                 99    ********************************
703A:  B9 81 C0  100            LDA   PHSON,Y
703D:  A9 56     101            LDA   #$56
703F:  20 A8 FC  102            JSR   MONWAIT
                 103   ********************************
                 104   *                              *
                 105   * TURN OFF PHASE (ALWAYS) &     *
                 106   * LOOP BACK TO CHECK ON NEW      *
                 107   * CURTRK-DESTRK RELATIONSHIP     *
                 108   *                              *
                 109   ********************************
7042:  B9 80 C0  110            LDA   PHSOFF,Y
7045:  4C 21 70  111            JMP   MAINLOOP
                 112   ********************************
                 113   *                              *
                 114   * ALL DONE... QUIT              *
                 115   *                              *
                 116   ********************************
7048:  A6 01     117   ALLDONE  LDX   SLOT
704A:  BD 88 C0  118            LDA   DISKOFF,X
704D:  60        119            RTS

--- END ASSEMBLY ---

TOTAL ERRORS: 00

78 BYTES OF OBJECT CODE
WERE GENERATED THIS ASSEMBLY.

]
```